

Pembuatan Model Transien Mesin Induksi Menggunakan PSPICE

Ayong Hiendro

Energy Conversion Laboratory
Department of Electrical Engineering, Engineering Faculty
Tanjungpura University.
E-mail : ayongh2000@yahoo.com

Abstract– For the investigation of transient in a complex system, a circuit based simulation program may be a skillful tool. PSPICE has large libraries of the electronic component's models, but only few models of electromechanical devices are available there. Therefore to generate a model of the induction machine is important. All non-electrical parameters must be replaced by electrical terms in the PSPICE model. Mechanical equations of the induction machine can be realized by either basic capacitance or inductance circuit topology. The complete model can be used to simulate the electrical and mechanical transient response of the induction machine.

Keywords– model, transient , induction machine, PSPICE

1. Pendahuluan

Pengoperasian mesin induksi pada industri merupakan suatu sistem mekatronik yang terdiri dari gabungan antara elektronik dan mekanik. Mesin induksi beroperasi pada kondisi transien ketika terjadi perubahan nilai amplitudo dan frekuensi dari besaran-besaran listrik atau kecepatan putaran mesin. Contoh dari operasi transien suatu mesin induksi adalah: mesin yang baru dimatikan langsung dihidupkan lagi, pembebanan secara tiba-tiba, tegangan listrik turun-naik, mesin dioperasikan melalui sumber listrik dengan konverter PWM.


Penelitian fenomena transien mesin induksi dapat dilakukan dengan membuat suatu model sebagai pengganti. Kebanyakan simulasi komputer dilakukan menggunakan bantuan program perangkat lunak MATLAB [1-3].

Akan tetapi sistem pengoperasian mesin induksi biasanya tidak bekerja sendirian, namun digabungkan dengan peralatan elektronik yang rumit seperti misalnya konverter elektronika daya. Untuk itu program simulasi yang berbasis rangkaian listrik (PSPICE) akan lebih memberikan kemudahan dan keunggulan.

Perangkat lunak PSPICE tidak menyediakan model jadi untuk mesin induksi, sehingga untuk melakukan simulasi komputer diperlukan suatu model mesin induksi yang tepat agar kinerja mesin dapat terwakilkan. Memang ada beberapa hasil penelitian yang telah membuat model mesin induksi berdasarkan PSPICE [4], namun tidak menjelaskan secara rinci model tersebut. Pada referensi [5] juga menulis tentang model mesin induksi dengan PSPICE, namun dalam melakukan

simulasi lebih cenderung model tersebut dijalankan menggunakan perangkat lunak GNUCAP karena di dalam model yang dibuat menggunakan komponen VCR (Voltage Controlled Resistor) untuk mendefinisikan kecepatan dan torka elektromagnetik dari mesin induksi. Tulisan ini menyajikan model untuk simulasi transien mesin induksi pada saat pengasutan (*starting*) dalam keadaan tanpa beban untuk mengamati respons kecepatan rotor dan torka elektromagnetik. Semua variabel magnetik (fluksi) dan variabel mekanik (kecepatan, torka, inersia) dari mesin induksi dimodelkan dalam bentuk rangkaian listrik menggunakan komponen listrik yang ada di dalam pustaka PSPICE. Selain itu diberikan juga contoh program yang dapat langsung dijalankan dengan menggunakan PSPICE, setelah semua nilai parameter mesin induksi diketahui.

2. Pemodelan Rangkaian Listrik untuk Mesin Induksi



Gambar 1. Diagram Mesin Induksi 3-Fasa untuk Model DQ

$$\begin{bmatrix} v_q \\ v_d \\ v_0 \end{bmatrix} = \frac{2}{3} \begin{bmatrix} \cos(\theta) & \cos(\theta - 2\pi/3) & \cos(\theta + 2\pi/3) \\ \sin(\theta) & \sin(\theta - 2\pi/3) & \sin(\theta + 2\pi/3) \\ 1/2 & 1/2 & 1/2 \end{bmatrix} \begin{bmatrix} v_a \\ v_b \\ v_c \end{bmatrix} \quad (1)$$

sedangkan kebalikan transformasi dari kerangka referensi dq ke kerangka referensi abc adalah:

$$\begin{bmatrix} v_a \\ v_b \\ v_c \end{bmatrix} = \frac{2}{3} \begin{bmatrix} \cos(\theta) & \sin(\theta) & 1 \\ \cos(\theta - 2\pi/3) & \sin(\theta - 2\pi/3) & 1 \\ \cos(\theta + 2\pi/3) & \sin(\theta + 2\pi/3) & 1 \end{bmatrix} \begin{bmatrix} v_q \\ v_d \\ v_0 \end{bmatrix} \quad (2)$$

Menggunakan persamaan tegangan v_a , v_b dan v_c (persamaan tegangan stator dan rotor) mesin induksi 3-fasa [6,7], kemudian ditransformasikan dengan persamaan (1), maka persamaan tegangan dalam kerangka referensi dq:

$$v_{qs} = r_s i_{qs} + \frac{d\lambda_{qs}}{dt} + \omega_e \lambda_{ds} \quad (3)$$

$$v_{qr} = r_r i_{qr} + \frac{d\lambda_{qr}}{dt} + \omega_r \lambda_{dr} \quad (4)$$

$$v_{ds} = r_s i_{ds} + \frac{d\lambda_{ds}}{dt} - \omega_e \lambda_{qs} \quad (5)$$

$$v_{dr} = r_r i_{dr} + \frac{d\lambda_{dr}}{dt} - \omega_r \lambda_{qr} \quad (6)$$

dan

$$\lambda_{qs} = L_s i_{qs} + L_m i_{qr} \quad (7)$$

$$\lambda_{ds} = L_s i_{ds} + L_m i_{dr} \quad (8)$$

$$\lambda_{qr} = L_m i_{qs} + L_r i_{qr} \quad (9)$$

$$\lambda_{dr} = L_m i_{ds} + L_r i_{dr} \quad (10)$$

$$L_s = L_{ls} + L_m \quad (11)$$

$$L_r = L_{lr} + L_m \quad (12)$$

$$\omega_r = \omega_e - \omega \quad (13)$$

di mana:

v_{qs} , v_{qr} = tegangan stator dan rotor sumbu-q, Volt

v_{ds} , v_{dr} = tegangan stator dan rotor sumbu-d, Volt

i_{qs} , i_{qr} = arus stator dan rotor sumbu-q, Ampere

i_{ds} , i_{dr} = arus stator dan rotor sumbu-d, Ampere

λ_{qs} , λ_{qr} = fluksi gandeng stator dan rotor sumbu-q, Wb.t

λ_{ds} , λ_{dr} = fluksi gandeng stator dan rotor sumbu-d, Wb.t

L_s , L_r = induktansi stator dan rotor, H

r_s , r_r = resistansi stator dan rotor, Ω

L_m = induktansi bersama, H


L_{ls} , L_{lr} = induktansi bocor (sendiri) stator dan rotor, H

ω_e = frekuensi listrik (eksitasi) stator, rad/s

ω_r = frekuensi listrik (slip) rotor, rad/s

ω = frekuensi sudut (kecepatan) rotor, rad/s

Persamaan (3 – 13) diterjemahkan ke dalam model rangkaian listrik menghasilkan rangkaian ekuivalen untuk mesin induksi 3-fasa seperti pada Gambar 1.



Gambar 2. Rangkaian Listrik Mesin Induksi

Pada rangkaian ekuivalen mesin induksi 3-fasa (Gambar 1), E_{ds} , E_{dr} , E_{qs} dan E_{qr} merupakan komponen sumber tegangan yang dikendalikan oleh tegangan lain ($E = \text{Voltage Controlled Dependent Voltage Source}$) yang terbentuk dari persamaan-persamaan:

$$E_{ds} = \omega_e \lambda_{qs} \quad (14)$$

$$E_{dr} = (\omega_e - \omega) \lambda_{qr} \quad (15)$$

$$E_{qs} = \omega_e \lambda_{ds} \quad (16)$$

$$E_{qr} = (\omega_e - \omega) \lambda_{dr} \quad (17)$$

Selanjutnya fluksi gandeng λ_{qs} , λ_{qr} , λ_{ds} , λ_{dr} masing-masing dapat dimodelkan sebagai suatu sumber tegangan yang terdiri dari dua buah sumber tegangan dikendalikan oleh arus ($H = \text{Current Controlled Dependent Voltage Source}$) yang terhubung secara seri, sehingga:

$$\lambda_{qs} = H_{qs1} + H_{qs2} \quad (18)$$

$$\lambda_{ds} = H_{ds1} + H_{ds2} \quad (19)$$

$$\lambda_{qr} = H_{qr1} + H_{qr2} \quad (20)$$

$$\lambda_{dr} = H_{dr1} + H_{dr2} \quad (21)$$

Sumber tegangan yang dikendalikan oleh arus pada persamaan (18 – 21) dapat didefinisikan berdasarkan persamaan (7 – 10) sebagai berikut:

$$H_{qs1} = L_s i_{qs} \quad (22)$$

$$H_{qs2} = L_m i_{qr} \quad (23)$$

$$H_{ds1} = L_s i_{ds} \quad (24)$$

$$H_{ds2} = L_m i_{dr} \quad (25)$$


$$H_{qr1} = L_m i_{qs} \quad (26)$$

$$H_{qr2} = L_r i_{qr} \quad (27)$$

$$H_{dr1} = L_m i_{ds} \quad (28)$$

$$H_{dr2} = L_r i_{dr} \quad (29)$$

Rangkaian listrik yang mewakili persamaan fluksi gandeng (18 -21) ditunjukkan pada Gambar 3.



Gambar 3. Rangkaian Listrik Fluksi Gandeng

Pada Gambar 3, komponen resistor r_x disisipkan ke dalam rangkaian sebagai pengukur tegangan antar terminal untuk mendapatkan nilai daripada λ_{qs} , λ_{qr} , λ_{ds} , λ_{dr} . Resistor yang disisipkan tersebut memiliki nilai yang cukup besar untuk menjadikannya sebagai Voltmeter.

Pada mesin induksi 3-fasa, bagian mekanik yaitu kecepatan rotor memiliki hubungan yang memenuhi persamaan:

$$\omega = \frac{1}{J_r} \int T_e dt \quad (30)$$

dan torka elektromagnetik adalah:

$$T_e = \frac{3}{2} \frac{P}{2} \frac{L_m}{L_r} (i_{qs} \lambda_{dr} - i_{ds} \lambda_{qr}) \quad (31)$$

Di mana:

J_r = momen inersia rotor, kg.m^2

T_e = torka elektromagnetik, N.m

P = jumlah kutub

Semua variabel dalam simulasi menggunakan PSPICE merupakan besaran listrik, sehingga variabel mekanik seperti kecepatan dan torka harus digantikan oleh variabel listrik. Persamaan mekanik (30) dapat dianalogikan menjadi persamaan listrik. Kalau dilihat dari sifat persamaan (30) ada dua cara untuk menggantikannya menjadi persamaan listrik, yaitu:


1. Menggunakan persamaan tegangan kapasitor:

$$v = \frac{1}{C} \int i dt \quad (32)$$

2. Menggunakan persamaan tegangan induktor:

$$v = L \frac{di}{dt} \quad (33)$$


Jika persamaan kecepatan mesin (30) dianalogikan dengan persamaan tegangan kapasitor (32), maka model yang dibuat akan menggantikan arus dengan torka, kapasitansi dengan inersia, serta tegangan dengan kecepatan.



Gambar 4. Rangkaian Listrik Kapasitor Pengganti Inersia

Pada Gambar 4, G_{Te} adalah sumber arus yang dikendalikan oleh tegangan ($G = \text{Voltage Controlled Dependent Current Source}$) yang nilainya adalah sesuai dengan persamaan (31). Nilai torka elektromagnetik yang dihasilkan oleh mesin induksi adalah arus yang mengalir di dalam rangkaian yaitu $i_{Te} = i(G_{Te})$. Sedangkan kecepatan rotor diwakilkan oleh hasil pengukuran tegangan pada terminal kapasitor inersia (C_j). Suatu resistansi yang cukup besar (r_j) disisipkan ke dalam rangkaian untuk mendapatkan tegangan tersebut.

Untuk persamaan kecepatan mesin (30) yang dianalogikan dengan persamaan tegangan induktor (33) maka model rangkaian listriknya seperti pada Gambar 5. Di sini arus menggantikan kecepatan, induktansi menggantikan inersia dan tegangan menggantikan torka. E_{Te} adalah sumber tegangan yang dikendalikan oleh tegangan lainnya yang nilainya sama dengan hasil perhitungan dari persamaan (31). Arus yang mengalir dalam rangkaian i_w menunjukkan besarnya kecepatan mesin induksi, dan untuk pengukuran arus perlu disisipkan suatu resistansi yang cukup kecil r_j seri terhadap induktansi inersia rangkaian.



Gambar 5. Rangkaian Listrik Induktor Pengganti Inersia

3. Pembuatan Program dengan PSPICE

Parameter mesin induksi yang dipergunakan adalah:

Tipe: tiga-fasa, 5.5kW, 6-kutub

$r_s = 1.05 \Omega$, $r_r = 0.754 \Omega$

$L_s = 256.6 \text{ mH}$, $L_r = 260.3 \text{ mH}$

$J_r = 0.2 \text{ kg.m}^2$

Dengan menggunakan kerangka referensi diam, maka nilai $\omega_e = 0$. Program PSPICE berdasarkan rangkaian listrik Gambar 1,2 dan 3 adalah:

```

*
*RANGKAIAN SUMBU D
*
VDS 1 0 SIN(0 310 50 0 0 90)
VXDS 1 1A DC 0
RLDS 1A 2 1.05
LLDS 2 3 3.6M IC=0
EDS 4 3 16 0 0
LMD 4 0 253M IC=0
EDR 4 5 26 0 1
LLDR 6 5 7.3M IC=0
RLDR 7A 6 0.754
VXDR 7 7A DC 0
VDR 7 0 DC 0
*
*RANGKAIAN SUMBU Q
*
VQS 8 0 SIN(0 310 50 0 0 0)
VXQS 8 8A DC 0
RLQS 8A 9 1.05
LLQS 9 10 3.6M IC=0
EQS 10 11 18 0 0
LMQ 11 0 253M IC=0
EQR 12 11 27 0 1
LLQR 13 12 7.3M IC=0
RLQR 14A 13 0.754
VXQR 14 14A DC 0
VQR 14 0 DC 0
*
*RANGKAIAN FLUKSI
*
*FLUKSI QS
*
HQS1 15 0 VXQS 257M
HQS2 16 15 VXQR 253M
RQS 16 0 1000
*
*FLUKSI DS
*
HDS1 17 0 VXDS 257M
HDS2 18 17 VXDR 253M
RDS 18 0 1000
*
*FLUKSI QR
*
HQR1 19 0 VXQS 253M
HQR2 20 19 VXQR 260M
RQR 20 0 1000
*
*FLUKSI DR
*
H1DR 21 0 VXDS 253M
H2DR 22 21 VXDR 260M
RDR 22 0 1000
*
*TORKA ELEKTRMAGNETIK (KAPASITOR)

```

```

E25 25 0 VALUE={I(VXQS)*V(22)}
E24 24 0 VALUE={I(VXDS)*V(20)}
GTE 0 23 VALUE={4.38*(V(25)-V(24))}
CJ 23 0 0.2 IC=0
RJ 23 0 1MEG
*KECEPATAN V(23)
*TORKA I(GTE)
*
*KECEPATAN (KAPASITOR)
EDRX 26 0 VALUE={(0-V(23))*V(20)}
EQRX 27 0 VALUE={(0-V(23))*V(22)}
*
.TRAN 1US 1600MS
.PROBE
.END

```

Apabila model rangkaian Gambar 4 diganti dengan model pada Gambar 5, maka program PSPICE untuk bagian ini adalah:

```

*TORKA ELEKTROMAGNETIK (INDUKTOR)
E25 25 0 VALUE={I(VXQS)*V(22)}
E24 24 0 VALUE={I(VXDS)*V(20)}
ETE 23 0 VALUE={4.38*(V(25)-V(24))}
VXTE 23 23A DC 0
LJ 23A 28 0.2 IC=0
RJ 28 0 1M
*TORKA V(23)
*KECEPATAN I(VXTE)
*
*KECEPATAN (INDUKTOR)
EDRX 26 0 VALUE={(0-I(VXTE))*V(20)}
EQRX 27 0 VALUE={(0-I(VXTE))*V(22)}

```

Parameter VXDS, VXDR, VXQS, VXQR dan VXTE yang disisipkan dalam penulisan program tidak mengubah fungsi program yang dibuat. Perlunya menambahkan parameter-parameter tersebut adalah sebagai pengukur arus yang mengalir di dalam rangkaian. Oleh karena itu selalu ditulis bernilai 0 Volt DC.


Selain itu di dalam program yang ditulis terdapat beberapa parameter lain yang tidak ada di dalam model rangkaian karena sudah termaktub di dalamnya, yaitu E24, E25, EDRX dan EQRX. Parameter E24 adalah terjemahan daripada hasil perkalian $(i_{ds} \lambda_{qr})$ dan E25

adalah $(i_{qs} \lambda_{dr})$. Hasil pengurangan (E25 – E24) nantinya merupakan bagian dari torka elektromagnetik (persamaan 31). Untuk model rangkaian Gambar 4 menjadi GTE dan model Gambar 5 menjadi ETE. Parameter EDRX adalah bagian dari pendefinisian untuk parameter EDR pada persamaan 15, sedangkan EQRX bagian dari definisi parameter EQR pada persamaan 17.

4. Hasil Simulasi Mesin Induksi


Hasil simulasi transien mesin induksi pada saat pengasutan (*starting*) tanpa beban ditunjukkan pada Gambar 6, 7 dan 8. Dengan menggunakan model rangkaian mekanik gambar 4, kecepatan rotor diukur dengan menggunakan parameter V(23) dan torka I(GTE). Sedangkan apabila model rangkaian mekanik

menggunakan Gambar 5, maka kecepatan rotor diukur menggunakan parameter $I(VXTE)$ dan torka $V(23)$.




Gambar 6. Arus Transien Stator

Kapasitas (*rating*) arus mesin induksi yang dipergunakan adalah 10A. Hasil simulasi menunjukkan pada saat starting (Gambar 6) arus transien mencapai 80A atau 8 kali arus *rating*-nya.. Lamanya waktu transien adalah 0.8 detik, dan setelah masa transien selesai, mesin induksi beroperasi kondisi mantap.



Gambar 7. Kecepatan Transien Rotor Saat Starting



Gambar 8. Torka Transien Saat Starting

Dalam periode transien, torka mesin beresilasi dengan amplitudo mencapai lebih dari dua kali lipat dari torka *rating*-nya. Setelah periode osilasi selesai nilai torka naik sebesar 40% di atas *rating*-nya dan kemudian turun menjadi nol pada keadaan mantap karena mesin diasut dalam keadaan tanpa beban.

5. Kesimpulan

1. Model transien mesin induksi yang dibuat berdasarkan persamaan matematik dan diterjemahkan menjadi model rangkaian listrik dapat digunakan untuk melakukan simulasi komputer menggunakan program perangkat lunak PSPICE.
2. Untuk dapat menggunakan PSPICE semua parameter non-listrik seperti parameter magnetik (fluksi) dan parameter mekanik (inersia, kecepatan, torka) harus dianalogikan menjadi parameter listrik.
3. Parameter inersia mesin induksi dapat dimodelkan dalam dua cara, yaitu menggunakan topologi rangkaian kapasitansi atau induktansi. Kedua model pengganti menghasilkan luaran (*output*) simulasi yang sama.

Referensi

- [1] Shi, K.L., Chan, T.F., Wong, Y.K., Ho, S.L., Modelling And Simulation Of The Three-Phase Induction Motor Using Simulink, International Journal Elect. Eng. Educ., Vol. 36, p. 163–172, Manchester U.P., 1999.
- [2] Kishore, A., Prasad, R.C., Karan, B.M, MATLAB SIMULINK Based DQ Modeling and Dynamic Characteristics of Three Phase Self Excited Induction Generator, Progress In Electromagnetics Research Symposium 2006, Cambridge, USA, March 26-29, p.312-316, 2006
- [3] Shandu, K.S, Pahwa, V., Simulation Study Of Three-Phase Induction Motor With Variations In Moment Of Inertia, ARPN Journal of Engineering and Applied Sciences, Vol. 4, No. 6 p.72-77, 2009
- [4] Lehtla, T., Joller, J., Lehtla, M., Laugis, Parameter Identification and Comparison of an Induction Motor Models, Power Electronics and Variable Speed Drives, 2000. Eighth International Conference on (IEE Conf. Publ. No. 475) p. 201-205, 2000
- [5] Rinkviciene, R., Petrovas, A., Modelling AC Induction Drive in PSPICE, Electronics and Electrical Engineering T.190, ISSN 1392-1215 No.1 (73) p.29-32, 2007
- [6] Pillay, P., Levin, V., Mathematical Models for Induction Machines, Industry Applications Conference, 1995. Thirtieth IAS Annual Meeting, IAS '95 Vol.1 p.606-616, 1995
- [7] Krause, P.C., Wasynczuk, O., Sudhoff, S.D., Analysis of Electric Machinery and Drive System 2nd Ed., Wiley Inter-Science, A John Wiley & Sons Inc. Pub., New York, 2002

Biography

Ayong Hiendro, received B.Eng Degree from Universitas Tanjungpura Pontianak and Master Degree from Institut Teknologi Bandung (ITB) in 2000. He is presently working at Department Electrical Engineering Universitas Tanjungpura Pontianak. His research interests include Electrical Machines, Power Quality and Non-Conventional Energy Systems.

